Fourier-Reihen

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Sept 07   Update 14.05.08
Web: http://haftendorn.uni-lueneburg.de   www.mathematik-verstehen.de

##

Dateiname fourier-reihen.mn

Angabe der Funktion fH in der Hauptperiode.
Definition von f als periodische Funktion.

```
Ta:=0:Ti:=3: Te:=4:
fH:=t--> piecewise([Ta<t and t<Ti,2*t],[ Ti<=t and
t<=Te,24-6*t]) :
f:=t->fH(frac(t/(Te-Ta))*(Te-Ta)):
```

```
plotfunc2d(fH(t), f(t), Scaling=Constrained, LegendVisible=FALSE)
```



Definitionen von Feldern für die Koeffizienten,
übliche Festlegungen T Periode, om =omega= Kreisfrequenz.

```
n:=10:
a:=array(0..10):     b:=array(1..10):
T:=4:
om:=2*PI/T:
```

Das Verschiebungsglied kann oft auch elementargeometrisch bestimmt werden.
Hier als Dreieck g=4, h=6 -> F=12 und 2/T=2/4=1/2

```
a[0]:=2/T*int(fH(t),t=Ta..Te)
```

$$6$$

So kann man nur die anderen Koeffizienten berechnen.

```
a[1]:=Simplify(2/T*int(fH(t)*cos(1*om*t),t=Ta..Te));
b[1]:=2/T*int(fH(t)*sin(1*om*t),t=Ta..Te);
```

$$-\frac{16}{\pi^2}$$

$$-\frac{16}{\pi^2}$$

Automatische Berechnung von n Schwingungen

Automatische Berechnung von n Schwingungen

```
for k from 1 to 10 do
    a[k]:=2/T*int(fH(t)*cos(k*om*t),t=Ta..Te);
    b[k]:=2/T*int(fH(t)*sin(k*om*t),t=Ta..Te);
end_for:
```

For-Schleifen haben keine Ausgaben. Ausgabe folgt. Vereinfachen muss extra gefordert werden.

```
[Simplify(a[k]),Simplify( b[k])] $ k=1..n
```

$$\left[-\frac{16}{\pi^2}, -\frac{16}{\pi^2}\right], \left[-\frac{8}{\pi^2}, 0\right], \left[-\frac{16}{9\cdot\pi^2}, \frac{16}{9\cdot\pi^2}\right], [0, 0], \left[-\frac{16}{25\cdot\pi^2}, -\frac{16}{25\cdot\pi^2}\right], \left[-\frac{8}{9\cdot\pi^2}, 0\right.$$

Wenn man mit so einem Werkzeug arbeitet, braucht man über die Gesetzäßigkeit in den Ausgaben nicht nachzudenken. Falls nötig, könnte man es schaffen.
Im Folgenden wird die Summe bis Ordnung n gebildet.

```
su:=(t,n)->a[0]/2+_plus((a[k]*cos(k*om*t)+b[k]*sin(k*om*
t)) $ k=1..n);
```

$$(t, n) \rightarrow \frac{a_0}{2} + \_plus(a_k \cdot \cos(k \cdot om \cdot t) + b_k \cdot \sin(k \cdot om \cdot t) \ \$ \ k = 1 \,..n)$$

Die Grundschwingung und ihre Achse ist deutlich.

```
plotfunc2d(fH(t),su(t,n) $
n=1..10,a[0]/2,LegendVisible=FALSE)
```



```
plotfunc2d(su(t,10),f(t), LegendVisible=FALSE)
```



Die Reihen mit 10 Doppeltermen ist von der gegebenen Funktion kaum zu unterscheiden. Der interaktiv genommene Ausschnitt zeigt Unterschiede bes. an der Ecken.